TD4 - Séries entières, matrices et déterminants

Nicolas Estibals*

14, 21 et 28 janvier 2009

Dans ce TD nous allons nous servir d'algèbre linéaire, la bibliothèque `linalg` de Maple peut être utile à cet effet. Elle apporte des fonctions utiles telles que : `matrix` (pour construire des matrices), `evalm` (évaluation de matrice), `eigenvals`, `eigenvects` (valeurs et vecteurs propres)...

Exercice 1 Une matrice inversible ? ou pas !

On se donne la matrice $A = \begin{pmatrix} 1 & a & 3a \\ -a & 1 & a \\ 3a & -a & 1 \end{pmatrix}, a \in \mathbb{C}$.

1. Trouvez une CNS sur a que A soit inversible.
2. Déterminez les sous-espaces propres et les valeurs propres.
3. Diagonaliser la matrice A.
4. Donner l'expression de générale de A^n pour n entier.

Exercice 2 Approcher le mal...

Nous allons chercher à calculer le logarithme népérien de 666.

1. Rappelez la série entière de $\ln(1+x)$ et son domaine de convergence.
2. Quelle est la somme de la série $\sum_{n=1}^{\infty} \frac{1}{n*2^n}$.
3. Donnez, sous forme de série, la différence entre $\ln(k+1)$ et $\ln(k)$.
4. Décomposez $\ln(666)$ en une expression n'utilisant que $\ln(2)$ et la formule précédente.
5. Donnez une approximation de $\ln(666)$ et vérifiez là grâce à Maple.
6. *(facultatif)* Estimez l'erreur que vous avez commise.

Exercice 3 Application linéaire

Soit $f : \mathbb{R}^3 \to \mathbb{R}^3$ tel que $f(x, y, z) = (3x - z, y - 2x, z - x - y)$.

1. Montrez que f est linéaire.
2. Écrivez sa matrice dans la base canonique.
3. Donnez le noyau, l'image et le rang de f.

Exercice 4 Racine

Soit $A = \begin{pmatrix} 125 & 234 & 0 & -234 \\ 0 & 1 & 0 & 0 \\ 0 & -248 & 125 & 0 \\ 0 & -7 & 0 & 8 \end{pmatrix}$, trouvez un B tel que $B^3 = A$.

*N'hésitez pas à m'écrire à l'adresse `Nicolas.Estibals@ens-lyon.fr`!
Vous pouvez retrouver les sujets sur `http://perso.ens-lyon.fr/nicolas.estibals/`.

TD4 : Séries entières, matrices et déterminants.

Nicolas Estibals (Nicolas.Estibals@ens-lyon.fr, http://perso.ens-lyon.fr/nicolas.estibals)

Exercice 1 : Matrice inversible ? ou pas !

```
> restart;
  with(linalg): # C'est un exercice d'algèbre linéaire, nous aurons
  surement besoin de l'une ou l'autre des fonctions de cette
  bibliothèque.
```

Warning, the protected names norm and trace have been redefined and unprotected

Question 1/

Être inversible, pour un matrice, est équivalent à être de déterminant non nul.

```
> A:=matrix(3,3,[1,a,3*a,-a,1,a,3*a,-a,1]);
  solve(det(A),a);
```

$$A := \begin{bmatrix} 1 & a & 3a \\ -a & 1 & a \\ 3a & -a & 1 \end{bmatrix}$$

$$1, \frac{1}{2}, \frac{-1}{3}$$

Ainsi A est inversible si et seulement si a est dans C – {-1/3,1/2,1}.

Question 2/

Il suffit d'utiliser la commande adaptée de la bibliothèque 'linalg'.

```
> eigenvects(A);
```

$$[-a+1, 1, \{[-1, -2, 1]\}], [3a+1, 1, \{[1, 0, 1]\}], [-2a+1, 1, \{[-1, -1, 1]\}]$$

Le résulats se lit ainsi : chaque liste correspond à un sous-espace propre et donne dans l'odre : la valeur propre associé, sa multiplicité et un ensemble de vecteurs propres engendrant le sous-espace correspondant.

Question 3/

Il nous suffit alors de poser la matrice de passage en juxtaposant les vecteurs propres. et de verifier le resultat. Nous verrons dans l'exercice 4 comment demander ceci directement à Maple.

```
> P:=matrix(3,3,[1,-1,1,0,-2,1,1,1,-1]);
  A=P&*diag(3*a+1,1-a,-2*a+1)&*P**(-1);
  evalm(%);
```

$$P := \begin{bmatrix} 1 & -1 & 1 \\ 0 & -2 & 1 \\ 1 & 1 & -1 \end{bmatrix}$$

$$A = \left(P \,\&^*\, \left(\begin{bmatrix} 3a+1 & 0 & 0 \\ 0 & -a+1 & 0 \\ 0 & 0 & -2a+1 \end{bmatrix} \right) \right) \&^* \left(\frac{1}{P} \right)$$

$$\begin{bmatrix} 1 & a & 3a \\ -a & 1 & a \\ 3a & -a & 1 \end{bmatrix} = \begin{bmatrix} 1 & a & 3a \\ -a & 1 & a \\ 3a & -a & 1 \end{bmatrix}$$

Question 4/

Il n'est pas possible de répondre directement à la question avec la matrice non diagonalisée, par contre il est aisée d'y répondre grace à la forme diagonale :

```
> A**n=evalm(P&*diag((3*a+1)**n,(1-a)**n,(1-2*a)**n)&*P**(-1));
```

$$A^n = \left[\left[\frac{1}{2}(3a+1)^n - \frac{1}{2}(-a+1)^n + (-2a+1)^n, (-a+1)^n - (-2a+1)^n, \frac{1}{2}(3a+1)^n + \frac{1}{2}(-a+1)^n - (-2a+1)^n \right], \right.$$
$$[-(-a+1)^n + (-2a+1)^n, 2(-a+1)^n - (-2a+1)^n, (-a+1)^n - (-2a+1)^n],$$
$$\left. \left[\frac{1}{2}(3a+1)^n + \frac{1}{2}(-a+1)^n - (-2a+1)^n, -(-a+1)^n + (-2a+1)^n, \frac{1}{2}(3a+1)^n - \frac{1}{2}(-a+1)^n + (-2a+1)^n \right] \right]$$

Exercice 2 : Approcher le mal...

```
> restart;
```

Question 1/

```
> u:=(n,t) -> (-1)**(n+1)*t**n/n:
  ln(1+t)=Sum(u(n,t),n=1..infinity); # pour t dans ]-1,1]
```

$$\ln(1+t) = \sum_{n=1}^{\infty} \frac{(-1)^{(n+1)} t^n}{n}$$

Question 2/

```
> Sum(1/(n*2**n),n=1..infinity)=sum(1/(n*2**n),n=1..infinity);
```

$$\sum_{n=1}^{\infty} \frac{1}{n\,2^n} = \ln(2)$$

Question 3/

Pour k>=2, k+1/k<2 et l'on peut ainsi se servir du dévellopement en série entière de la question 1/ pour ln((k+1)/k).

```
> ln(k+1)=ln(k)+ln((k+1)/k);
  ln(k+1)=ln(k)+simplify(Sum(u(n,1/k),n=1..infinity));
```

$$\ln(k+1) = \ln(k) + \ln\left(\frac{k+1}{k}\right)$$

$$\ln(k+1) = \ln(k) + \left(\sum_{n=1}^{\infty}\left(-\frac{(-k)^{(-n)}}{n}\right)\right)$$

Question 4/

On va utiliser la formule précédente et le dévellopement en facteur premier de 66 pour obtenir le résultat.

```
> ifactor(666);
  ln(666)=ln(2)+2*ln(3)+ln(37);
```

$$(2)\,(3)^2\,(37)$$

$$\ln(666) = \ln(2) + 2\ln(3) + \ln(37)$$

```
> ln(37)=ln(36)+Sum(u(n,1/36),n=1..infinity);
  ifactor(36);
  'ln'(36)=2*ln(2)+2*ln(3); # Notez l'utilisation de ' ' pour demander à
  maple de ne pas évaluer la fonction.
```

$$\ln(37) = 2\ln(6) + \left(\sum_{n=1}^{\infty}\frac{(-1)^{(n+1)}\left(\frac{1}{36}\right)^n}{n}\right)$$

$$(2)^2\,(3)^2$$

$$2\ln(6) = 2\ln(2) + 2\ln(3)$$

```
> ln(666)=
  7*ln(2)+4*Sum(u(n,1/2),n=1..infinity)+Sum(u(n,1/36),n=1..infinity);
```

$$\ln(666) = 7\ln(2) + 4\left(\sum_{n=1}^{\infty}\frac{(-1)^{(n+1)}\left(\frac{1}{2}\right)^n}{n}\right) + \left(\sum_{n=1}^{\infty}\frac{(-1)^{(n+1)}\left(\frac{1}{36}\right)^n}{n}\right)$$

Question 5/

La question précédente nous donne une série qui converge vers ln(666), pour en obtenir une approximation il suffit alors d'estimer la somme sur ses 50 premiers termes par exemple.

```
> v:=(n,t)->7*(1/(n*2**n)) + 4*u(n,1/2) +u(n,1/36);
  ln(666)=Sum(v(n,t), n=1..infinity);
  evalf(sum(v(n,t),n=1..50));
  evalf(ln(666));
```

$$v := (n, t) \to \frac{7}{n\,2^n} + 4\,u\left(n, \frac{1}{2}\right) + u\left(n, \frac{1}{36}\right)$$

$$\ln(666) = \sum_{n=1}^{\infty}\left(\frac{7}{n\,2^n} + \frac{4\,(-1)^{(n+1)}\left(\frac{1}{2}\right)^n}{n} + \frac{(-1)^{(n+1)}\left(\frac{1}{36}\right)^n}{n}\right)$$

$$6.501289671$$

$$6.501289671$$

Question 6/

Pour estimez l'erreur commise il suffit de majorer le reste de la série.

```
> v(n,t) < 12/n/2**n;
  Sum(12/n/2**n,n=51..infinity) = evalf(sum(12/n/2**n,n=51..infinity));
```

$$\frac{7}{n\,2^n} + \frac{4\,(-1)^{(n+1)}\left(\frac{1}{2}\right)^n}{n} + \frac{(-1)^{(n+1)}\left(\frac{1}{36}\right)^n}{n} < \frac{12}{n\,2^n}$$

$$\sum_{n=51}^{\infty}\frac{12}{n\,2^n} = 2.051080453727401658209721 0^{-16}$$

Ainsi il y a au moins 15 chiffres après la virgule de correct, ce que nous avons déjà pu vérifions ci-dessous. Notez que si vous utilisez la série de la question 1/ pour évaluer ln(2), vous aurez beaucoup moint de chiffres corrects, en effet cette série converge lentement pour t=1.

```
> Digits:=20;
  evalf(sum(v(n,t),n=1..50));
  evalf(ln(666));
```

$$Digits := 20$$

6.5012896705403889936

6.5012896705403891366

Exercice 3 ; Application linéaire

```
> restart;
  with(linalg):
```

Warning, the protected names norm and trace have been redefined and unprotected

Question 1/

Il suffit pour cela de vérifier les deux axiomes de linéarité pour les applications dans un espace vectoriel.

```
> f:=(x,y,z)->(3*x-z, y-2*x,z-x-y);
  'f'(k*x,k*y,k*z)-k*'f'(x,y,z)     = simplify(f(k*x,k*y,k*z)-k*f(x,y,z))
  'f'(x+u,y+v,z+w)-'f'(x,y,z)-'f'(u,v,w) =
  simplify(f(x+u,y+v,z+w)-f(x,y,z)-f(u,v,w));
```

$$f := (x, y, z) \to 3\,x - z,\, y - 2\,x,\, z - x - y$$

$$f(k\,x, k\,y, k\,z) - k\,f(x, y, z) = 0$$

$$f(x + u, y + v, z + w) - f(x, y, z) + (-3\,u + w, -v + 2\,u, -w + u + v) = 0$$

Question 2/

Il suffit de l'écrire puis de vérifier quel'on ne s'est pas trompé.

```
> M:=matrix(3,3,[3,0,-1,-2,1,0,-1,-1,1]);
  evalm(M&*matrix(3,1,[x,y,z]));
```

$$M := \begin{bmatrix} 3 & 0 & -1 \\ -2 & 1 & 0 \\ -1 & -1 & 1 \end{bmatrix}$$

$$\begin{bmatrix} 3\,x - z \\ y - 2\,x \\ z - x - y \end{bmatrix}$$

Question 3/

La bibliothèque 'linalg' fournit toutes les commande nécessaire pour répondre directement à ces questions sur la matrice de f.

```
> # Pour le noyau
  kernel(M);
  # Pour l'image
  colspace(M);
  # Et enfin pour le rang
  rank(M);
```

$$\{[1, 2, 3]\}$$

$$\{[0, 1, -1], [1, 0, -1]\}$$

$$2$$

Exercice 4 : Racine

```
> restart;
  with(linalg):
```

Warning, the protected names norm and trace have been redefined and unprotected

```
> # Définissons d'abord A
  A:=matrix(4,4,[125,234,0,-234,0,1,0,0,0,-248,125,0,0,-7,0,8]);
```

$$A := \begin{bmatrix} 125 & 234 & 0 & -234 \\ 0 & 1 & 0 & 0 \\ 0 & -248 & 125 & 0 \\ 0 & -7 & 0 & 8 \end{bmatrix}$$

Pour répondre à la question il faut diagonaliser la matrice, il est alors aisée de prendre la "racine cubique" de cette matrice diagonale et de trouver un B.

```
> # 'linalg' fournit la commande 'jordan' qui permet de mettre un matric
  sous sa forme de Jordan, c'est à dire de la diagonaliser.
  jordan(A,'P');
  print(P); # P est la matrice de passage.
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 8 & 0 & 0 \\ 0 & 0 & 125 & 0 \\ 0 & 0 & 0 & 125 \end{bmatrix}$$

$$\begin{bmatrix} 0 & -2 & \frac{248}{7} & \frac{234}{7} \\ 1 & 0 & 0 & 0 \\ 2 & 0 & \frac{-262}{7} & \frac{-248}{7} \\ 1 & -1 & 0 & 0 \end{bmatrix}$$

Pour trouver un B il suffit alors de faire :

```
> B:=evalm(P&*diag(1,2,5,5)&*P**(-1));
```

$$B := \begin{bmatrix} 5 & 6 & 0 & -6 \\ 0 & 1 & 0 & 0 \\ 0 & -8 & 5 & 0 \\ 0 & -1 & 0 & 2 \end{bmatrix}$$

Vérifions le résultat :

```
> evalm(A-B**3);
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$